SAS Codes for our Algorithm:

```
/*The program calculates the test statistic -2 Log Likelihood
(chi_sq) and the corresponding p-value of the proposed test. The
program calculates the estimate S of the unstructured variance
covariance matrix Omega in Dental Data set A. In this program q=2,
the # of variables and p=3, the # of repeated measures in each
variable. It also calculates the estimates of V and Sigma of
covariance structure V@sigma.*/

options nocenter ls=80 ps=50 nodate nonumber;
data a; infile 'c:\zuloA.dat';
input sub x1-x9;

proc iml symbolsize=3036;
use a; read all var {x1 x2 x4 x5 x7 x8}
into gp;
read all var {x1 x4 x7} into gp1; read all var {x2 x5 x8}
into gp2;

pi=3.14159265;  converge=0.000001;
n=nrow(gp); q=2; p=3;
one={1,1,1}; jj1={1,0,0}; jj2={0,1,0}; jj3={0,0,1};

j0=j(q,q,0); j1=I(q)||j0||j0; j2=j0||I(q)||j0; j3=j0||j0||I(q);

/* gpbar is  the means of Dental Data set A.*/ gpbar=gp[+, ]/n;
gp1bar=gp1[+, ]/n; gp2bar=gp2[+, ]/n;

w=j(p*q,p*q,0); do j=1 to n;
   use a; read all var {x1 x2 x4 x5 x7 x8} into subb where (sub=j);
   w=w+t(subb-gpbar)*(subb-gpbar);
end;

gp1barm=repeat(gp1bar,n,1); gp1a=gp1-gp1barm; gp1ass=gp1a[##, ];
sq_gp1assm= repeat(sqrt(gp1ass),n,1); gp1an=gp1a/ sq_gp1assm;
vi1=t(gp1an)*gp1an;

gp2barm=repeat(gp2bar,n,1); gp2a=gp2-gp2barm; gp2ass=gp2a[##, ];
sq_gp2assm= repeat(sqrt(gp2ass),n,1); gp2an=gp2a/ sq_gp2assm;
vi2=t(gp2an)*gp2an;
vi=(vi1+vi2)/2;

/* Calculating the initial estimate Ve of V. rest is the pooled
estimate of rho*/

rest= (vi[+,+]-trace(vi))/(p*(p-1)); print rest;
ve=(1-rest)*I(p)+rest*one*t(one);

/* Calculating the mle's  mlv and mlsig of V and Sigma.*/
mlsige=j(q,q,0); iter=0; do until (maxab<converge);
   ive=inv(ve);
   v11=t(jj1)*ive*jj1; v12=t(jj1)*ive*jj2; v13=t(jj1)*ive*jj3;
   v21=t(jj2)*ive*jj1; v22=t(jj2)*ive*jj2; v23=t(jj2)*ive*jj3;
   v31=t(jj3)*ive*jj1; v32=t(jj3)*ive*jj2; v33=t(jj3)*ive*jj3;
```

```
   sigma_e=j(q,q,0);
   do j=1 to n;
      use a;
      read all var {x1 x2 x4 x5 x7 x8} into subb where (sub=j);
      subba=subb-gpbar;
      sub1=j1*t(subba); sub2=j2*t(subba); sub3=j3*t(subba);
      sigma_e=sigma_e+v11*((sub1)*t(sub1))+v21*((sub1)*t(sub2))+
              v31*((sub1)*t(sub3))+
              v12*((sub2)*t(sub1))+v22*((sub2)*t(sub2))+
              v32*((sub2)*t(sub3))+
              v13*((sub3)*t(sub1))+v23*((sub3)*t(sub2))+
              v33*((sub3)*t(sub3));
   end;

   mlsig=sigma_e/(n*p);
   absig=abs(trace(mlsige-mlsig));
   mlsige=mlsig;
   imlsig=inv(mlsig);
   k3=trace((I(p)@imlsig)*w);
   k4=trace(((one*t(one))@imlsig)*w);

   /* Solving the Cubic Equation (Abramowitz, M. and Stegun)*/

   s=p-1; ko=n*q*s*p;
   pp=(ko-s*ko+k3*s**2-s*k4)/(s*ko); qq=(2*s*k3-ko)/(s*ko); rr=(k3-k4)/(s*ko);
   aa=(1/3)*(3*qq-pp**2); bb=(1/27)*(2*pp**3 -9*pp*qq +27*rr);
   discrim=(bb**2)/4+(aa**3)/27;
   s1=((bb**2)/4+(aa**3)/27)**0.5;
   s2=(-bb/2+ s1)**(1/3); ar=-bb/2-s1;
   if ar <0 then
      do;
         ar1=-ar; ar2=ar1**(1/3); s3=-ar2;
      end;
   else
      do;
         s3=ar**(1/3);
      end;
   ro2=s2+s3-pp/3;
   mlv=(1-ro2)*I(p)+ro2*one*t(one);
   ve=mlv;
   iter=iter+1;
   abr1=abs(rest-ro2);
   rest=ro2;
   maxab=max(absig//abr1);
end;
print iter ro2 mlsig;
detw=det(w); log_detw=log(detw);
a=-0.5*trace(inv(ve@mlsig)*w);  b=-0.5*n*p*q;
detsig=(det(mlsig));  detve=det(ve);
a1=(detve**(-q*n/2));
b1=(detsig**(-p*n/2));
b2=(n**(n*p*q/2));
c=(2*pi)**(-n*p*q/2);
de=0.5*n*p*q;
```

```
tempho=-2*(  log(c)+ log(a1)+log(b1) ) +2*(-a); temph1=-2*(
log(c)+ ((-n/2)* log_detw) +log(b2) ) +2*de; temp=tempho-temph1;
print tempho temph1 temp;

chi_sq=tempho-temph1;
df=(p*q*(p*q+1)/2)-(q*(q+1)/2)-1;
p_val=1-probchi(chi_sq, df);
print df chi_sq p_val;
run;
```

SAS Codes using *Proc Mixed*:

```
/* Here we fit the linear fixed effects model under the null hypothesis
Omega= UN @ CS for the Dental Data set A (Timm, 1980, Table 7.2)
data taking only the first two variables. As a by-product we get "-2 Log
Likelihood" in the output.*/

options nocenter ls=80 ps=50 nodate nonumber;
data a;
infile 'c:\zuloA.dat';
input id x1-x9;

/* The data are converted to univariate form. The variable time representing
the repeated measures and the variable m_var representing the two
characteristics is also declared as a class variable.*/

data b; set a;
subj=_n_;
x=x1; m_var='var1'; time=1; output;
x=x2; m_var='var2'; time=1; output;
x=x4; m_var='var1'; time=2; output;
x=x5; m_var='var2'; time=2; output;
x=x7; m_var='var1'; time=3; output;
x=x8; m_var='var2'; time=3; output;
drop x1-x9;
proc mixed data=b method=ml covtest;
classes subj m_var time;
model x=m_var time m_var*time;
repeated m_var time/type= un@cs subject=subj;
Title'-2 Log Likelihood under the Null hypothesis Ho';
run;

/* Here we fit the model under the alternative  hypothesis Omega= UN
for the Dental Data set A taking only the first two
variables. As a byproduct we get -2 Log Likelihood in the output.

Subj represents the individuals and tt represents the time. All
values of y1,...,y6 are stacked one below the other as y with
corresponding levels of the variable time appropriately identified.*/

data c; set a;
array t{6} x1 x2 x4 x5 x7 x8;
subj+1;
```

```
do tt=1 to 6;
x=t{tt};
output;
end;
drop x1-x8;
proc mixed data=c method=ml covtest;
classes subj tt;
model x=tt;
repeated /type= un subject=subj;
Title'-2 Log Likelihood under the Alternative Hypothesis H1';
run;
```

Zulo Data Set A:

1	117	59	10.5	117.5	59	16.5	118.5	60	16.5
2	109	60	30.5	110.5	61.5	30.5	111	61.5	30.5
3	117	60	23.5	120	61.5	23.5	120.5	62	23.5
4	112	67.5	33	126	70.5	32	127	71.5	32.5
5	116	61.5	24.5	118.5	62.5	24.5	119.5	63.5	24.5
6	123	65.5	22	126	61.5	22	127	67.5	22
7	130.5	68.5	33	132	69.5	32.5	134.5	71	32
8	126.5	69	20	128.5	71	20	130.5	73	20
9	113	58	25	116.5	59	25	118	60.5	24.5